JOURNAL OF THEORETICAL
AND APPLIED MECHANICS
55, 3, pp. 787-800, Warsaw 2017
DOI: 10.15632/jtam-pl.55.3.787

FINITE ELEMENT IMPLEMENTATION OF SLIGHTLY COMPRESSIBLE AND INCOMPRESSIBLE FIRST INVARIANT-BASED HYPERELASTICITY: THEORY, CODING, EXEMPLARY PROBLEMS

Cyprian Suchocki
Warsaw University of Technology, Department of Mechanics and Armament Technology, Warsaw, Poland
e-mail: c.suchocki@imik.wip.pw.edu.pl

The present study is concerned with the finite element (FE) implementation of slightly compressible hyperelastic material models. A class of constitutive equations is considered where the isochoric potential functions are based on the first invariant of the right Cauchy-Green (C-G) deformation tensor. Special attention is paid to the most recently developed model formulations. The incremental form of hyperelasticity and its numerical implementation into both commercial and non-commercial FE software are discussed. A Fortran 77 UMAT code is attached which allows for a simple implementation of arbitrary first invariant-based constitutive models into Abaqus and Salome-Meca FE packages. Several exemplary problems are considered.

Keywords: hyperelasticity, finite element method, UMAT, elasticity tensor

1. Introduction

The hyperelastic constitutive equations are nowadays available in every advanced FE program. However, the material libraries of FE software usually include only a number of standard hyperelastic models such as: neo-Hooke, Mooney-Rivlin, Ogden or Yeoh. Less celebrated or newly developed constitutive models can be implemented into a FE program by taking advantage of a proper user subroutine. The FE package Abaqus provides three user subroutines which allow one to define a custom hyperelastic model, i.e. UHYPER (for isotropic hyperelastic materials), UANISOHYPER (for anisotropic hyperelastic materials) and UMAT (a general purpose subroutine which can be utilized for implementing any kind of constitutive equation), cf. Hibbit *et al.* (2008). Due to the method of FE implementation used for slightly compressible hyperelasticity in Abaqus, it is not recommended to utilize the subroutine UHYPER for all kinds of finite elements (cf. Jemioło, 2002). Thus, in the case of slightly compressible hyperelastic materials, i.e. the materials with decoupled volumetric and isochoric responses, the subroutine UMAT might be preferred. Both UHYPER and UANISOHYPER subroutines can be utilized to define nonlinear viscoelastic models based on the viscoelasticity theory used by Abaqus. Alternatively, a proper option allows one to simulate the Mullins effect in a hyperelastic material defined by the aforementioned subroutines[1]. On the other hand, the subroutine UMAT is a much more powerful tool which enables one to define an arbitrary constitutive theory, including those based on hyperelasticity such as nonlinear viscoelasticity (e.g. Suchocki 2013) or growth models (e.g. Young *et al.*, 2010), so that the user is not limited by the built-in options of Abaqus.

The subroutine UMAT is a Fortran 77 code which is called during every iteration of the Newton-Raphson numerical procedure to calculate components of the stress tensor and the material Jacobian which is also reffered to as tangent modulus or (in the case of elastic materials)

[1]The nonlinear viscoelasticity and the Mullins effect must be used separately as Abaqus does not allow for combining these behaviors.

the elasticity tensor, cf. Hibbit *et al.* (2008). The material Jacobian may be defined either in an approximate or (if possible) an analytical form, which is usually very difficult to determine. The approximate material Jacobians always worsen the rate of convergence of the numerical calculations. It was demonstrated by Stein and Sagar (2008) that for the neo-Hooke hyperelastic model, the quadratic rate of convergence[2] is obtained only when the analytical material Jacobian is used. The utilization of the approximate material Jacobians resulted in worsening the convergence rate and, in the case of some of the considered problems and finite element types, it caused lack of convergence. Thus, it is always recommended to use an analytical material Jacobian whenever it is available.

In this study, the FE implementation of slightly compressible isotropic hyperelastic constitutive models that are not included in any of the commercial and non-commercial CAE packages is discussed. The stored energy functions that are based on the first invariant of the isochoric right C-G tensor are considered. The focus is on the recently developed models for polymeric materials (Gent, 1996; Jemioło, 2002; Lopez-Pamies, 2010, da Silva Soares, 2008; Khajehsaeid *et al.*, 2013) and on some model formulations used in soft tissue biomechanics (Demiray, 1972; Demiray *et al.*, 1988). The general framework for deriving an analytical material Jacobian is presented. A subroutine UMAT is attached allowing for using the newly developed exponential-logarithmic model (Khajehsaeid *et al.*, 2013) in both Abaqus and Salome-Meca FE packages. The code structure is universal so that any other first invariant-based slightly compressible or incompressible hyperelastic model can be easily implemented by simply changing the expressions for the stored energy derivatives. A number of exemplary problems were solved for selected energy potentials. The presented UMAT code can be upgraded to define nonlinear viscoelastic, elastoplastic, viscoplastic or other behavior using arbitrary constitutive theory.

2. Slightly compressible hyperelastic materials

In the following derivations, the multiplicative split of the deformation gradient tensor into the volumetric and isochoric component is utilized (e.g. Jemioło, 2016), i.e.

$$\mathbf{F} = \mathbf{F}_{vol}\overline{\mathbf{F}} \qquad \mathbf{F}_{vol} = J^{\frac{1}{3}}\mathbf{1} \qquad \overline{\mathbf{F}} = J^{-\frac{1}{3}}\mathbf{F} \qquad \overline{\mathbf{C}} = \overline{\mathbf{F}}^{\mathrm{T}}\overline{\mathbf{F}} = J^{-\frac{2}{3}}\mathbf{C} \tag{2.1}$$

where $J = \det\mathbf{F}$ and $\overline{\mathbf{C}}$ is the isochoric right C-G tensor with the following set of algebraic invariants

$$\bar{I}_1 = \operatorname{tr}\overline{\mathbf{C}} \qquad \bar{I}_2 = \frac{1}{2}\Big((\operatorname{tr}\overline{\mathbf{C}})^2 - \operatorname{tr}\overline{\mathbf{C}}^2\Big) \qquad \bar{I}_3 = \det\overline{\mathbf{C}} = 1 \tag{2.2}$$

In the case of slightly compressible hyperelastic materials, the stored energy function is considered to be the sum of the volumetric contribution U and the isochoric part $\overline{W}$, thus

$$W(\mathbf{C}) = U(J) + \overline{W}(\bar{I}_1, \bar{I}_2) \qquad \mathbf{S} = 2\frac{\partial W}{\partial \mathbf{C}}\Big|_{\mathbf{C}=\mathbf{C}^{\mathrm{T}}} \tag{2.3}$$

where the most general form of the constitutive equation is given by Eq. $(2.3)_2$[3]. After substituting Eq. $(2.3)_1$ into Eq. $(2.3)_2$, the decoupled form of the constitutive equation is found

$$\mathbf{S} = Jp\mathbf{C}^{-1} + J^{-\frac{2}{3}}\,\mathrm{DEV}\,\big[\overline{\mathbf{S}}\big] \qquad p = \frac{\partial U}{\partial J} \qquad \overline{\mathbf{S}} = 2\frac{\partial \overline{W}}{\partial \overline{\mathbf{C}}}\Big|_{\overline{\mathbf{C}}=\overline{\mathbf{C}}^{\mathrm{T}}} \tag{2.4}$$

with $\mathrm{DEV}\,[\bullet] = [\bullet] - \frac{1}{3}\Big([\bullet]\cdot\overline{\mathbf{C}}\Big)\overline{\mathbf{C}}^{-1}$ being a deviator in the reference configuration.

[2]The quadratic convergence means that the error at the current iteration is proportional to the square of the error from the previous iteration.

[3]The adopted notation emphasizes the fact that symmetrization is carried out after calculating a derivative.

3. Material Jacobian tensor

Taking a directional derivative of Eq. $(2.4)_1$ with respect to $\mathbf{C}$, an incremental constitutive relation is found, see e.g. Jemioło and Gajewski (2014)

$$\Delta\mathbf{S} = \boldsymbol{\mathcal{C}} \cdot \frac{1}{2}\Delta\mathbf{C} \qquad \boldsymbol{\mathcal{C}} = 2\frac{\partial\mathbf{S}}{\partial\mathbf{C}}\Big|_{\mathbf{C}=\mathbf{C}^{\mathrm{T}}} = 4\frac{\partial^2 W}{\partial\mathbf{C}\otimes\partial\mathbf{C}}\Big|_{\mathbf{C}=\mathbf{C}^{\mathrm{T}}} \qquad \boldsymbol{\mathcal{C}} = \boldsymbol{\mathcal{C}}^{vol} + \boldsymbol{\mathcal{C}}^{iso} \tag{3.1}$$

Assuming $U = U(J)$ and $\overline{W} = \overline{W}(\bar{I}_1)$, the expressions for the volumetric and the isochoric parts of the elasticity tensor can be derived

$$\begin{aligned}
\boldsymbol{\mathcal{C}}^{vol} &= J\frac{\partial U}{\partial J}\Big(\mathbf{C}^{-1}\otimes\mathbf{C}^{-1} - 2\mathbf{I}_{\mathbf{C}^{-1}}\Big) + J^2\frac{\partial^2 U}{\partial J^2}\mathbf{C}^{-1}\otimes\mathbf{C}^{-1} \\
\boldsymbol{\mathcal{C}}^{iso} &= -\frac{4}{3}J^{-\frac{2}{3}}\frac{\partial\overline{W}}{\partial\bar{I}_1}\Big[\mathbf{1}\otimes\mathbf{C}^{-1} + \mathbf{C}^{-1}\otimes\mathbf{1} - I_1\Big(\mathbf{I}_{\mathbf{C}^{-1}} + \frac{1}{3}\mathbf{C}^{-1}\otimes\mathbf{C}^{-1}\Big)\Big] + J^{-\frac{4}{3}}\overline{\boldsymbol{\mathcal{C}}}_{\overline{W}} \\
\overline{\boldsymbol{\mathcal{C}}}_{\overline{W}} &= 4\frac{\partial^2\overline{W}}{\partial\bar{I}_1^2}\Big[\mathbf{1}\otimes\mathbf{1} - \frac{1}{3}I_1(\mathbf{1}\otimes\mathbf{C}^{-1} + \mathbf{C}^{-1}\otimes\mathbf{1}) + \frac{1}{9}I_1^2\mathbf{C}^{-1}\otimes\mathbf{C}^{-1}\Big]
\end{aligned} \tag{3.2}$$

where

$$\mathbf{I}_{\mathbf{C}^{-1}} = \frac{1}{2}\Big[(\mathbf{C}^{-1}\otimes\mathbf{C}^{-1})^{\overset{(2,3)}{\mathrm{T}}} + (\mathbf{C}^{-1}\otimes\mathbf{C}^{-1})^{\overset{(2,4)}{\mathrm{T}}}\Big] = \frac{1}{2}(C^{-1}_{IK}C^{-1}_{JL} + C^{-1}_{IL}C^{-1}_{JK})\mathbf{E}_I\otimes\mathbf{E}_J\otimes\mathbf{E}_K\otimes\mathbf{E}_L$$

is the fourth order identity tensor in the reference configuration with the Cartesian base $\{\mathbf{E}_K\}$ $(K = 1,2,3)$[4], see e.g. Suchocki (2011).

The incremental constitutive law given by Eq. $(3.1)_1$ can be transformed into a form relating the incremental Oldroyd (convected) rate of the Kirchhoff stress to the increment of the strain rate tensor, i.e.

$$\mathcal{L}_v\boldsymbol{\tau} = \Delta\boldsymbol{\tau} - \Delta\mathbf{L}\boldsymbol{\tau} - \boldsymbol{\tau}\Delta\mathbf{L}^{\mathrm{T}} = \boldsymbol{\mathcal{C}}^{\tau c}\cdot\Delta\mathbf{D} \tag{3.3}$$

where $\Delta\mathbf{L} = \Delta\mathbf{F}\mathbf{F}^{-1}$ is the increment of the velocity gradient, whereas $\boldsymbol{\mathcal{C}}^{\tau c}$ is the pushed-forward form of the material Jacobian

$$\boldsymbol{\mathcal{C}}^{\tau c} = F_{iP}F_{jQ}F_{kR}F_{lS}\mathcal{C}_{PQRS}\mathbf{e}_i\otimes\mathbf{e}_j\otimes\mathbf{e}_k\otimes\mathbf{e}_l \tag{3.4}$$

with $\{\mathbf{e}_k\}$ $(k = 1,2,3)$ being the Cartesian base in the current configuration. The elasticity tensor takes the following form

$$\begin{aligned}
\boldsymbol{\mathcal{C}}^{\tau c} &= \frac{4}{3}\frac{\partial\overline{W}}{\partial\bar{I}_1}\Big[\bar{I}_1\Big(\mathbf{I} - \frac{1}{3}\mathbf{1}\otimes\mathbf{1}\Big) - (\mathbf{1}\otimes\,\mathrm{dev}\,(\overline{\mathbf{B}}) + \,\mathrm{dev}\,(\overline{\mathbf{B}})\otimes\mathbf{1})\Big] + 4\frac{\partial^2\overline{W}}{\partial\bar{I}_1^2}\,\mathrm{dev}\,(\overline{\mathbf{B}})\otimes\,\mathrm{dev}\,(\overline{\mathbf{B}}) \\
&\quad + J\Big[\Big(\frac{\partial U}{\partial J} + J\frac{\partial^2 U}{\partial J^2}\Big)\mathbf{1}\otimes\mathbf{1} - 2\frac{\partial U}{\partial J}\mathbf{I}\Big]
\end{aligned} \tag{3.5}$$

where

$$\mathbf{I} = \mathbf{1}\diamond\mathbf{1} = \frac{1}{2}\Big[(\mathbf{1}\otimes\mathbf{1})^{\overset{(2,3)}{\mathrm{T}}} + (\mathbf{1}\otimes\mathbf{1})^{\overset{(2,4)}{\mathrm{T}}}\Big] = \frac{1}{2}(\delta_{ik}\delta_{jl} + \delta_{il}\delta_{jk})\mathbf{e}_i\otimes\mathbf{e}_j\otimes\mathbf{e}_k\otimes\mathbf{e}_l$$

and $\mathrm{dev}\,[\bullet] = [\bullet] - \frac{1}{3}([\bullet]\cdot\mathbf{1})\mathbf{1}$.

[4]The following notation is used: $[\bullet]^{\overset{(\mu,\nu)}{\mathrm{T}}} = \Big([\bullet]_{ijkl}\mathbf{e}_i\otimes\underbrace{\mathbf{e}_j}_{\mu}\otimes\mathbf{e}_k\otimes\underbrace{\mathbf{e}_l}_{\nu}\Big)^{\overset{(\mu,\nu)}{\mathrm{T}}} = [\bullet]_{ijkl}\mathbf{e}_i\otimes\underbrace{\mathbf{e}_l}_{\mu}\otimes\mathbf{e}_k\otimes\underbrace{\mathbf{e}_j}_{\nu}$.

The FE software Abaqus utilizes the incremental constitutive equation written in terms of the incremental Zaremba-Jaumann rate of the Kirchhoff stress (cf. Hibbit *et al.* 2008), i.e.

$$\boldsymbol{\tau}^{\nabla} = \Delta\boldsymbol{\tau} - \Delta\mathbf{W}\boldsymbol{\tau} - \boldsymbol{\tau}\Delta\mathbf{W}^{\mathrm{T}} = J\boldsymbol{\mathcal{C}}^{MJ} \cdot \Delta\mathbf{D} \tag{3.6}$$

where, respectively

$$\boldsymbol{\mathcal{C}}^{MJ} = \frac{1}{J}(\boldsymbol{\mathcal{C}}^{\tau c} + \mathbf{1}\diamond\boldsymbol{\tau} + \boldsymbol{\tau}\diamond\mathbf{1}) \qquad \boldsymbol{\tau} = Jp\mathbf{1} + 2\frac{\partial\overline{W}}{\partial\bar{I}_1}\,\mathrm{dev}\,(\overline{\mathbf{B}}) \tag{3.7}$$

and

$$\mathbf{1}\diamond\boldsymbol{\tau} = \frac{1}{2}\Big[(\mathbf{1}\otimes\boldsymbol{\tau})^{\overset{(2,3)}{\mathrm{T}}} + (\mathbf{1}\otimes\boldsymbol{\tau})^{\overset{(2,4)}{\mathrm{T}}}\Big] \qquad \boldsymbol{\tau}\diamond\mathbf{1} = \frac{1}{2}\Big[(\boldsymbol{\tau}\otimes\mathbf{1})^{\overset{(2,3)}{\mathrm{T}}} + (\boldsymbol{\tau}\otimes\mathbf{1})^{\overset{(2,4)}{\mathrm{T}}}\Big]$$

and

$$\Delta\mathbf{W} = \frac{1}{2}(\Delta\mathbf{L} - \Delta\mathbf{L}^{\mathrm{T}}) \qquad \Delta\mathbf{D} = \frac{1}{2}(\Delta\mathbf{L} + \Delta\mathbf{L}^{\mathrm{T}}) \tag{3.8}$$

The fourth order tensor $\boldsymbol{\mathcal{C}}^{MJ}$ is the material Jacobian which should be coded in the subroutine UMAT. For the considered class of hyperelastic materials, it takes the form

$$\begin{aligned}\boldsymbol{\mathcal{C}}^{MJ} &= \frac{2}{J}\frac{\partial\overline{W}}{\partial\bar{I}_1}\Big[\mathbf{1}\diamond\,\mathrm{dev}\,(\overline{\mathbf{B}}) + \mathrm{dev}\,(\overline{\mathbf{B}})\diamond\mathbf{1} + \frac{2}{3}\bar{I}_1\Big(\mathbf{I} - \frac{1}{3}\mathbf{1}\otimes\mathbf{1}\Big)\\ &\quad - \frac{2}{3}(\mathbf{1}\otimes\,\mathrm{dev}\,(\overline{\mathbf{B}}) + \mathrm{dev}\,(\overline{\mathbf{B}})\otimes\mathbf{1})\Big] + \frac{4}{J}\frac{\partial^2\overline{W}}{\partial\bar{I}_1^2}\,\mathrm{dev}\,(\overline{\mathbf{B}})\otimes\,\mathrm{dev}\,(\overline{\mathbf{B}}) + \Big(\frac{\partial U}{\partial J} + J\frac{\partial^2 U}{\partial J^2}\Big)\mathbf{1}\otimes\mathbf{1}\end{aligned} \tag{3.9}$$

4. Finite element implementation

4.1. General

In Fig. 1, the interaction of the subroutine UMAT with the Abaqus package is illustrated for the Newton-Raphson iterative procedure during a single time increment (cf. Hibbit *et al.* 2008).



Fig. 1. Flow chart for the interaction of Abaqus and UMAT

The subroutine UMAT calculates the components of Cauchy stress and material Jacobian for each Gauss integration point. These quantities are subsequently used by Abaqus to form up the element stiffness matrix. Finally, the global stiffness matrix is assembled by Abaqus using the element stiffness matrices. The user subroutines used in other FE packages to define custom constitutive equations are integrated with the remainder of the program in a similar way and play the same role.

4.2. Variables and dimensions

In the following table, the meaning of the variables used in the Fortran 77 code has been explained. The dimensions of array variables have been specified in proper indices. The lengthy definitions of the auxiliary variables have been skipped.

Number of direct stress components	NDI
Number of shear stress components	NSHR
Array of material constants	PROPS(I)
Deformation gradient tensor $\mathbf{F}_{3\times3}$	DFGRD1(I,J)
Jacobian determinant	DET
Isochoric deformation gradient matrix $\overline{\mathbf{F}}_{3\times3}$	DISTGR(I,J)
Isochoric Left C-G deformation tensor matrix $\overline{\mathbf{B}}_{6\times1}$	BBAR(I)
Trace of $\overline{\mathbf{B}}$ divided by 3	TRBBAR
First partial derivative $\partial_J U$	DUDJ
Second partial derivative $\partial^2_{J^2} U$	DDUDDJ
First partial derivative $\partial_{\bar{I}_1}\overline{W}$	DWDI1
Second partial derivative $\partial^2_{\bar{I}_1^2}\overline{W}$	DDWDDI1
Cauchy stress tensor matrix $\boldsymbol{\sigma}_{6\times1}$	STRESS(I)
Material Jacobian matrix $\boldsymbol{\mathcal{C}}^{MJ}_{6\times6}$	DDSDDE(I,J)
Auxiliary variables	EK, PR, SCALE, TERM1, TERM2, TERM3

According to the rule adopted in Abaqus, the column matrix components $1, 2, \ldots, 6$ correspond to the scalar components of the second order tensor: 11, 22, 33, 12, 13, 23, respectively.

4.3. User subroutine UMAT

Algorithm for the implementation in ABAQUS

Input data: $\mathbf{F}_{3\times3}$ (DFGRD1), NDI, NSHR

1. Calculate Jacobian determinant J (DET)

$$J = \det \mathbf{F}_{3\times3}$$

2. Calculate isochoric deformation gradient $\overline{\mathbf{F}}_{3\times3}$ (DISTGR)

$$\overline{\mathbf{F}}_{3\times3} = J^{-\frac{1}{3}}\mathbf{F}_{3\times3}$$

3. Calculate left C-G deformation tensor $\overline{\mathbf{B}}_{6\times1}$ (BBAR)

$$\overline{\mathbf{B}}_{3\times3} = \overline{\mathbf{F}}_{3\times3}\overline{\mathbf{F}}^{\mathrm{T}}_{3\times3} \qquad \overline{\mathbf{B}}_{6\times1} = \{\overline{B}_{11}\ \overline{B}_{22}\ \overline{B}_{33}\ \overline{B}_{12}\ \overline{B}_{13}\ \overline{B}_{23}\}^{\mathrm{T}}$$

4. Calculate Cauchy stress matrix $\boldsymbol{\sigma}_{6\times1}$ (STRESS)
5. Calculate Material Jacobian matrix $\boldsymbol{\mathcal{C}}^{MJ}_{6\times6}$ (DDSDDE).

4.4. Coding in Fortran 77

```
      SUBROUTINE UMAT(STRESS,STATEV,DDSDDE,SSE,SPD,SCD,
     1 RPL,DDSDDT,DRPLDE,DRPLDT,STRAN,DSTRAN,
     2 TIME,DTIME,TEMP,DTEMP,PREDEF,DPRED,MATERL,NDI,NSHR,NTENS,
     3 NSTATV,PROPS,NPROPS,COORDS,DROT,PNEWDT,CELENT,
     4 DFGRD0,DFGRD1,NOEL,NPT,KSLAY,KSPT,KSTEP,KINC)
!
      INCLUDE 'ABA_PARAM.INC'
!
      CHARACTER*8 MATERL
      DIMENSION STRESS(NTENS),STATEV(NSTATV),
     1 DDSDDE(NTENS,NTENS),DDSDDT(NTENS),DRPLDE(NTENS),
     2 STRAN(NTENS),DSTRAN(NTENS),DFGRD0(3,3),DFGRD1(3,3),
     3 TIME(2),PREDEF(1),DPRED(1),PROPS(NPROPS),COORDS(3),DROT(3,3)
!

!     LOCAL ARRAYS
! ----------------------------------------------------------------
!     BBAR   - DEVIATORIC RIGHT CAUCHY-GREEN TENSOR
!     DISTGR - DEVIATORIC DEFORMATION GRADIENT (DISTORTION TENSOR)
! ----------------------------------------------------------------
!
      REAL*8 BBAR,DISTGR
      DIMENSION BBAR(6),DISTGR(3,3)
!
      PARAMETER(ZERO=0.D0, ONE=1.D0, TWO=2.D0, THREE=3.D0, FOUR=4.D0)
!
! ----------------------------------------------------------------
!   UMAT FOR COMPRESSIBLE EXPONENTIAL-LOGARITHMIC HYPERELASTICITY
!
!                     WARSAW UNIVERSITY OF TECHNOLOGY
!                       CYPRIAN SUCHOCKI, JULY 2015
!
!                     CANNOT BE USED FOR PLANE STRESS
! ----------------------------------------------------------------
!     PROPS(1) - A
!     PROPS(2) - A1
!     PROPS(3) - B
!     PROPS(4) - D1
! ----------------------------------------------------------------
      REAL*8 A, A1, B, D1, TERM1, TERM2, TERM3, DUDJ, DDUDDJ,
     1  DWDI1, DDWDDI1, TRBBAR, DET, SCALE
!
!     ELASTIC PROPERTIES
!
      A=0.195
      A1=0.018 ! originally a
      B=0.22
      D1=0.000000033
!
!     JACOBIAN AND DISTORTION TENSOR
!
      DET=DFGRD1(1, 1)*DFGRD1(2, 2)*DFGRD1(3, 3)
     1   -DFGRD1(1, 2)*DFGRD1(2, 1)*DFGRD1(3, 3)
      IF(NSHR.EQ.3) THEN
        DET=DET+DFGRD1(1, 2)*DFGRD1(2, 3)*DFGRD1(3, 1)
     1          +DFGRD1(1, 3)*DFGRD1(3, 2)*DFGRD1(2, 1)
     2          -DFGRD1(1, 3)*DFGRD1(3,1)*DFGRD1(2, 2)
     3          -DFGRD1(2, 3)*DFGRD1(3, 2)*DFGRD1(1, 1)
      END IF
      SCALE=DET**(-ONE/THREE)
      DO K1=1, 3
        DO K2=1, 3
          DISTGR(K2, K1)=SCALE*DFGRD1(K2, K1)
        END DO
      END DO
!
!     CALCULATE LEFT CAUCHY-GREEN TENSOR
!
      BBAR(1)=DISTGR(1, 1)**2+DISTGR(1, 2)**2+DISTGR(1, 3)**2
```

```
      BBAR(2)=DISTGR(2, 1)**2+DISTGR(2, 2)**2+DISTGR(2, 3)**2
      BBAR(3)=DISTGR(3, 3)**2+DISTGR(3, 1)**2+DISTGR(3, 2)**2
      BBAR(4)=DISTGR(1, 1)*DISTGR(2, 1)+DISTGR(1, 2)*DISTGR(2, 2)
     1        +DISTGR(1, 3)*DISTGR(2, 3)
      IF(NSHR.EQ.3) THEN
        BBAR(5)=DISTGR(1, 1)*DISTGR(3, 1)+DISTGR(1, 2)*DISTGR(3, 2)
     1          +DISTGR(1, 3)*DISTGR(3, 3)
        BBAR(6)=DISTGR(2, 1)*DISTGR(3, 1)+DISTGR(2, 2)*DISTGR(3, 2)
     1          +DISTGR(2, 3)*DISTGR(3, 3)
      END IF
!
!     CALCULATE THE STRESS
!
      TRBBAR=(BBAR(1)+BBAR(2)+BBAR(3))/THREE
      DUDJ=2/D1*(DET-ONE)
      DDUDDJ=2/D1
      DWDI1=A*(exp(A1*(THREE*TRBBAR-THREE))
     1 -B*log(THREE*TRBBAR-TWO))
      DDWDDI1=A*(A1*exp(A1*(THREE*TRBBAR-THREE))
     1 -B*(THREE*TRBBAR-TWO)**(-ONE))
      TERM1=-FOUR/(THREE*DET)*DWDI1
      TERM2=FOUR/DET*DDWDDI1
      TERM3=DET*DDUDDJ

      CALL CALCSTRESS(BBAR,TRBBAR,DET,DUDJ,DWDI1,NDI,NSHR,
     1 STRESS)
!
!     CALCULATE THE STIFFNESS
!
      CALL CALCTANGENT(DDSDDE,STRESS,BBAR,TRBBAR,DUDJ,
     1 DWDI1,DDWDDI1,TERM1,TERM2,TERM3,DET,NTENS,NSHR)
!
      RETURN
      END
! ----------------------------------------------------------------

      SUBROUTINE CALCSTRESS(BBAR,TRBBAR,DET,DUDJ,DWDI1,NDI,NSHR,
     1 STRESS)

      REAL*8 BBAR,TRBBAR,DET,DUDJ,DWDI1,STRESS
      DIMENSION BBAR(6),STRESS(6)

      PARAMETER(TWO=2.D0)

      INTEGER NDI,NSHR,K1

      DO K1=1,NDI
        STRESS(K1)=TWO/DET*DWDI1*( BBAR(K1)-TRBBAR)+DUDJ
      END DO
      DO K1=NDI+1,NDI+NSHR
        STRESS(K1)=TWO/DET*DWDI1*BBAR(K1)
      END DO

      RETURN
      END

! ----------------------------------------------------------------

      SUBROUTINE CALCTANGENT(DDSDDE,STRESS,BBAR,TRBBAR,DUDJ,
     1 DWDI1,DDWDDI1,TERM1,TERM2,TERM3,DET,NTENS,NSHR)

      REAL*8 DDSDDE,STRESS,BBAR,TRBBAR,DUDJ,DWDI1,DDWDDI1,
     1 TERM1,TERM2,TERM3,DET
      DIMENSION DDSDDE(6,6),STRESS(6),BBAR(6)

      INTEGER NTENS,NSHR,K1,K2

      PARAMETER(TWO=2.D0, THREE=3.D0, FOUR=4.D0)

      DDSDDE(1, 1)=-DUDJ+TERM3+TWO*TERM1*(BBAR(1)-TWO*TRBBAR)+
     1  TERM2*(BBAR(1)**TWO+TRBBAR*(-TWO*BBAR(1)+TRBBAR))+
```

```
     2  TWO*STRESS(1)
      DDSDDE(2, 2)=-DUDJ+TERM3+TWO*TERM1*(BBAR(2)-TWO*TRBBAR)+
     1  TERM2*(BBAR(2)**TWO+TRBBAR*(-TWO*BBAR(2)+TRBBAR))+
     2  TWO*STRESS(2)
      DDSDDE(3, 3)=-DUDJ+TERM3+TWO*TERM1*(BBAR(3)-TWO*TRBBAR)+
     1  TERM2*(BBAR(3)**TWO+TRBBAR*(-TWO*BBAR(3)+TRBBAR))+
     2  TWO*STRESS(3)
      DDSDDE(1, 2)=DUDJ+TERM3+TERM1*(BBAR(1)+BBAR(2)-TRBBAR)+
     1  TERM2*(BBAR(1)*BBAR(2)-TRBBAR*(BBAR(1)+BBAR(2))+
     2  TRBBAR**TWO)
      DDSDDE(1, 3)=DUDJ+TERM3+TERM1*(BBAR(1)+BBAR(3)-TRBBAR)+
     1  TERM2*(BBAR(1)*BBAR(3)-TRBBAR*(BBAR(1)+BBAR(3))+
     2  TRBBAR**TWO)
      DDSDDE(2, 3)=DUDJ+TERM3+TERM1*(BBAR(2)+BBAR(3)-TRBBAR)+
     1  TERM2*(BBAR(2)*BBAR(3)-TRBBAR*(BBAR(2)+BBAR(3))
     2  +TRBBAR**TWO)
      DDSDDE(1, 4)=FOUR/DET*BBAR(4)*(-DWDI1/THREE+
     1  DDWDDI1*(BBAR(1)-TRBBAR))+STRESS(4)
      DDSDDE(2, 4)=FOUR/DET*BBAR(4)*(-DWDI1/THREE+
     1  DDWDDI1*(BBAR(2)-TRBBAR))+STRESS(4)
      DDSDDE(3, 4)=FOUR/DET*BBAR(4)*(-DWDI1/THREE+
     1  DDWDDI1*(BBAR(3)-TRBBAR))
      DDSDDE(4, 4)=-DUDJ+TWO/DET*(TRBBAR*DWDI1+
     1  TWO*DDWDDI1*BBAR(4)**TWO)+(STRESS(1)+STRESS(2))/TWO
      IF(NSHR.EQ.3) THEN
        DDSDDE(1, 5)=FOUR/DET*BBAR(5)*(-DWDI1/THREE+
     1    DDWDDI1*(BBAR(1)-TRBBAR))+STRESS(5)
        DDSDDE(2, 5)=FOUR/DET*BBAR(5)*(-DWDI1/THREE+
     1    DDWDDI1*(BBAR(2)-TRBBAR))
        DDSDDE(3, 5)=FOUR/DET*BBAR(5)*(-DWDI1/THREE+
     1    DDWDDI1*(BBAR(3)-TRBBAR))+STRESS(5)
        DDSDDE(1, 6)=FOUR/DET*BBAR(6)*(-DWDI1/THREE+
     1    DDWDDI1*(BBAR(1)-TRBBAR))
        DDSDDE(2, 6)=FOUR/DET*BBAR(6)*(-DWDI1/THREE+
     1    DDWDDI1*(BBAR(2)-TRBBAR))+STRESS(6)
        DDSDDE(3, 6)=FOUR/DET*BBAR(6)*(-DWDI1/THREE+
     1    DDWDDI1*(BBAR(3)-TRBBAR))+STRESS(6)
        DDSDDE(5, 5)=-DUDJ+TWO/DET*(TRBBAR*DWDI1+
     1    TWO*DDWDDI1*BBAR(5)**TWO)+(STRESS(1)+STRESS(3))/TWO
        DDSDDE(6, 6)=-DUDJ+TWO/DET*(TRBBAR*DWDI1+
     1    TWO*DDWDDI1*BBAR(6)**TWO)+(STRESS(2)+STRESS(3))/TWO
        DDSDDE(4,5)=TERM2*BBAR(4)*BBAR(5)+STRESS(6)/TWO
        DDSDDE(4,6)=TERM2*BBAR(4)*BBAR(6)+STRESS(5)/TWO
        DDSDDE(5,6)=TERM2*BBAR(5)*BBAR(6)+STRESS(4)/TWO
      END IF
      DO K1=1, NTENS
        DO K2=1, K1-1
          DDSDDE(K1, K2)=DDSDDE(K2, K1)
        END DO
      END DO

      RETURN
      END
```

5. Exemplary problems

A number of exemplary FE simulations have been prepared in order to verify the performance of the developed UMAT code. Seven different types of the isochoric stored energy potential $\overline{W}(\bar{I}_1)$ and two types of the volumetric function $U(J)$ have been tested (see Tables 1 and 2). Two different approaches were used in order to simulate the material near incompressibility, i.e. the penalty method and the hybrid formulation (e.g. Liu *et al.* 1994). The results obtained for the material near incompressibility in the case of homogenous deformations were compared to the analytical solutions available in the fully incompressible case.

Table 1. Material parameter values

Material	Constitutive parameters	Units
Jemioło (2002) – Lopez-Pamies (2010)	$\mu_1 = 2.228$ $\mu_2 = 1.919$ $\alpha_1 = 0.6$ [-],$\alpha_2 = -68.73$	[MPa] [MPa] [–]
Gent (1996)	$\mu = 0.27$ $J_m = 85.91$	[MPa] [–]
Khajehsaeid *et al.* (2013)	$A = 0.195$ $a = 0.018$ $b = 0.22$	[MPa] [–] [–]
Demiray (1971)	$c = 0.2$ $\beta = 16$	[MPa] [–]
Demiray *et al.* (1988)	$\alpha = 10.74$E-10 $\beta = 7.548$E-9 $c = 1.17$	[MPa] [MPa] [–]
Da Silva Soares (2008)	$\mu_1 = 17.999$ $\mu_2 = 0.17047$ $a = 477.28$	[MPa] [MPa] [–]
Knowles (1977)	$\mu = 264.069$ $b = 54.19$ $n = 0.2554$	[MPa] [–] [–]

5.1. Simple tension

In the case of uniaxial tension of an incompressible rectangular block (Fig. 2) along the X_1-direction, the deformation is defined by the set of equations

$$x_1 = \lambda_1 X_1 \qquad x_2 = \lambda_1^{-\frac{1}{2}} X_2 \qquad x_3 = \lambda_1^{-\frac{1}{2}} X_3 \tag{5.1}$$

where the stretch ratio $\lambda_1 > 1$ and $J = 1$ is assumed. It follows that

$$I_1 = \lambda_1^2 + \frac{2}{\lambda_1} \qquad W = W(I_1) \tag{5.2}$$

which yields an equation for the axial component of the Lagrange stress

$$T_{11} = 2\frac{\partial W}{\partial I_1}\Big(\lambda_1 - \frac{1}{\lambda_1^2}\Big) \tag{5.3}$$

The analytical Eq. (5.3) was used to verify the results of FE calculations. In numerical simulation, a 15 mm×15 mm×15 mm block was undergoing a uniaxial tension (Fig. 2). In the first approach, a single C3D8[5] element was used with the material near incompressibility being enforced by using the penalty method. The penalty parameter $D_1 = 33\text{E-9}\,\text{MPa}^{-1}$. In the second approach, a hybrid element C3D8H was utilized. The comparison of the numerical results and the analytical solution for the incompressible material can be seen in Fig. 3. The FE simulations were later repeated for the block meshed with 125 elements which produced exactly the same results.

[5]Cubic, three-dimensional, 8 nodes.

Table 2. Exemplary isochoric and volumetric stored-energy functions and their derivatives

Material	Energy potential $\overline{W}(\bar{I}_1)$	1st derivative $\partial\overline{W}/\partial\bar{I}_1$	2nd derivative $\partial^2\overline{W}/\partial\bar{I}_1^2$
Jemioło (2002) – Lopez-Pamies (2010)	$\sum_{r=1}^{M}\frac{3^{1-\alpha_r}}{2\alpha_r}\mu_r\left(\bar{I}_1^{\alpha_r}-3^{\alpha_r}\right)$	$\sum_{r=1}^{M}\frac{3^{1-\alpha_r}}{2}\mu_r\bar{I}_1^{\alpha_r-1}$	$\sum_{r=1}^{M}\frac{3^{1-\alpha_r}}{2}\mu_r(\alpha_r-1)\bar{I}_1^{\alpha_r-2}$
Gent (1996)	$-\frac{\mu J_m}{2}\ln\left(1-\frac{\bar{I}_1-3}{J_m}\right)$	$\frac{\mu}{2}\left(1-\frac{\bar{I}_1-3}{J_m}\right)^{-1}$	$\frac{\mu}{2J_m}\left(1-\frac{\bar{I}_1-3}{J_m}\right)^{-2}$
Khajehsaeid *et al.* (2013)	$A\Big[\frac{1}{a}\mathrm{e}^{a(\bar{I}_1-3)}-\frac{1}{a}-b$ $+b(\bar{I}_1-2)\left(1-\ln(\bar{I}_1-2)\right)\Big]$	$A\left[\mathrm{e}^{a(\bar{I}_1-3)}-b\ln(\bar{I}_1-2)\right]$	$A\left[a\mathrm{e}^{a(\bar{I}_1-3)}-b(\bar{I}_1-2)^{-1}\right]$
Demiray (1971)	$c\left(\mathrm{e}^{\beta(\bar{I}_1-3)}-1\right)$	$c\beta\mathrm{e}^{\beta(\bar{I}_1-3)}$	$c\beta^2\mathrm{e}^{\beta(\bar{I}_1-3)}$
Demiray *et al.* (1988)	$\frac{\alpha}{4}(\bar{I}_1-3)^2+\frac{\beta}{4c}\left[\mathrm{e}^{c(\bar{I}_1-3)^2}-1\right]$	$\frac{1}{2}(\bar{I}_1-3)\left[\alpha+\beta\mathrm{e}^{c(\bar{I}_1-3)^2}\right]$	$\frac{1}{2}\left\{\alpha+\beta\mathrm{e}^{c(\bar{I}_1-3)^2}[1+2c(\bar{I}_1-3)^2]\right\}$
Da Silva Soares (2008)	$\mu_1\mathrm{e}^{-(\bar{I}_1-3)}(\bar{I}_1-3)$ $+\mu_2\ln[1+a(\bar{I}_1-3)]$	$\mu_1\mathrm{e}^{-(\bar{I}_1-3)}(4-\bar{I}_1)$ $+\mu_2 a[1+a(\bar{I}_1-3)]^{-1}$	$-\mu_1\mathrm{e}^{-(\bar{I}_1-3)}(5-\bar{I}_1)$ $-\mu_2 a^2[1+a(\bar{I}_1-3)]^{-2}$
Knowles (1977)	$\frac{\mu}{2b}\left\{\left[1+\frac{b}{n}(\bar{I}_1-3)\right]^n-1\right\}$	$\frac{\mu}{2}\left[1+\frac{b}{n}(\bar{I}_1-3)\right]^{n-1}$	$\frac{\mu b(n-1)}{2n}\left[1+\frac{b}{n}(\bar{I}_1-3)\right]^{n-2}$

Material	Energy potential $U(J)$	1st derivative $\partial U/\partial J$	2nd derivative $\partial^2 U/\partial J^2$
Sussman and Bathe (1987)	$\frac{1}{D_1}(J-1)^2$	$\frac{2}{D_1}(J-1)$	$\frac{2}{D_1}$
Simo and Taylor (1982)	$\frac{1}{D_1}[(J-1)^2+(\ln J)^2]$	$\frac{2}{D_1}\left(J+\frac{\ln J}{J}-1\right)$	$\frac{2}{D_1}\left[1+\frac{1}{J^2}(1-\ln J)\right]$



Fig. 2. Uniaxial defomation of a single element: (a) distribution of the displacement, (b) boundary conditions



Fig. 3. Uniaxial tension for various hyperelastic models; comparison of analytical and FE results: (a) Demiray (1972), (b) Demiray *et al.* (1988), (c) Exp-Ln, (d) Gent, (e) Jemioło–Lopez-Pamies, (f) Da Silva Soares

5.2. Simple shear

In the case of simple shear of an incompressible rectangular block in the $X_1 - X_2$ plane (Fig. 4), the deformation is defined by the set of equations

$$x_1 = X_1 + \gamma X_2 \qquad x_2 = X_2 \qquad x_3 = X_3 \tag{5.4}$$

where $\gamma > 0$. The first invariant of the right C-G tensor is given as

$$I_1 = \gamma^2 + 3 \tag{5.5}$$

which yields the following components of the Lagrange stress tensor

$$\mathbf{T}_{3\times 3} = \frac{2}{3}\frac{\partial W}{\partial I_1}\begin{bmatrix} -\gamma^2 & 3\gamma & 0 \\ \gamma(\gamma^2+3) & -\gamma^2 & 0 \\ 0 & 0 & -\gamma^2 \end{bmatrix} \tag{5.6}$$



Fig. 4. Shear deformation of a single element: (a) distribution of the displacement, (b) boundary conditions

The analytical formula for T_{12} given by Eq. (5.6) was utilized to verify the results of FE calculations. In numerical simulation, a 15 mm×15 mm×15 mm block was undergoing a simple shear (Fig. 4). Again, the analysis was carried out using the penalty method with a single C3D8 element ($D_1 = 33\text{E-}9\,\text{MPa}^{-1}$) and was subsequently repeated for a hybrid element C3D8H. The comparison of the numerical results and the analytical solution for the incompressible material can be seen in Fig. 5. The FE simulations were later performed for the block meshed with 125 elements with exactly the same results.



Fig. 5. Simple shear for various hyperelastic models; comparison of analytical and FE results: (a) Demiray (1972), (b) Demiray *et al.* (1988), (c) Exp-Ln, (d) Gent, (e) Jemioło–Lopez-Pamies, (f) Da Silva Soares

6. Conclusions

In this paper, the FE implementation of slightly compressible, first invariant-based, isotropic hyperelastic constitutive equations is discussed. Special attention is paid to the newly developed models for polymers and some of the stored energy functions used in the soft tissue biomechanics. A user subroutine UMAT code is attached, which enables the implementation of the aforementioned models into Abaqus and Salome-Meca FE packages. The performance of this code has been verified using some exemplary problems and an excellent agreement was found with the analytical solutions. It should be emphasized that the stress-stretch (or stress-amount of shear) relation which yields from the potential function developed by Demiray *et al.* (1988) is characterized by a very flat slope in the small strain domain (cf. Figs. 3b and 5b). Thus, for this particular model, a considerably small strain increment should be used initially in order to avoid convergence problems. The presented UMAT code can be further modified in order to define any constitutive theory that would be an extension of the slightly compressible, first invariant-based hyperelasticity.

References

1. Da Silva Soares J.F., 2008, Constitutive modeling for biodegradable polymers for application in endovascular stents, PhD thesis, Texas A&M University
2. Demiray H., 1972, A note on the elasticity of soft biological tissues, *Journal of Biomechanics*, **5**, 3, 309-311
3. Demiray H., Weizsäcker H.W., Pascale K., Erbay H.A., 1988, A stress-strain relation for a rat abdominal aorta, *Journal of Biomechanics*, **21**, 5, 369-374
4. Gent A.N., 1996, A new constitutive relation for rubber, *Rubber Chemistry and Technology*, **69**, 59-61
5. Hibbit B., Karlsson B., Sorensen P., 2008, *ABAQUS Theory Manual*, Hibbit, Karlsson & Sorensen Inc.
6. Jemioło S., 2002, A study on the hyperelastic properties of isotropic materials (in Polish), *Scientific Surveys of Warsaw University of Technology*, **140**, Warsaw University of Technology Publishing House, Warsaw
7. Jemioło S., 2016, Constitutive relations of hyperelasticity (in Polish), *Studies in the Field of Engineering*, **94**, The Committee on Civil Engineering and Hydroengineering of the Polish Academy of Sciences, Warsaw
8. Jemioło S., Gajewski M., 2014, Hyperelastoplasticity (in Polish), *Monographs of the Department of Strength of Materials, Theory of Elasticity and Plasticity*, **4**, Warsaw University of Technology Publishing House, Warsaw
9. Khajehsaeid H., Arghavani J., Naghdabadi R., 2013, A hyperelastic constitutive model for rubber-like materials, *European Journal of Mechanics A/Solids*, **38**, 2, 144-151
10. Knowles J.K., 1977, The finite anti-plane shear field near the tip of a crack for a class of incompressible elastic solids, *International Journal of Fracture*, **13**, 5, 611-639
11. Liu C.H., Hofstetter G., Mang H.A., 1994, 3D finite element analysis of rubber-like materials at finite strains, *Engineering Computations*, **11**, 111-128
12. Lopez-Pamies O., 2010, A new I_1-based hyperelastic model for rubber elastic materials, *Comptes Rendus Mecanique*, **338**, 1, 3-11
13. Simo J.C., Taylor R.L., 1982, Penalty function formulations for incompressible nonlinear elastostatics, *Computer Methods in Applied Mechanics and Engineering*, **35**, 107-118
14. Stein E., Sagar G., 2008, Convergence behavior of 3D finite elements for neo-Hookean material, *Engineering Computations: International Journal for Computer-Aided Engineering and Software*, **25**, 3, 220-232
15. Suchocki C., 2011, A finite element implementation of Knowles stored-energy function: theory, coding and applications, *Archive of Mechanical Engineering*, **58**, 3, 319-346
16. Suchocki C., 2013, A quasi-linear viscoelastic rheological model for thermoplastics and resins, *Journal of Theoretical and Applied Mechanics*, **51**, 1, 117-129

17. Sussman T., Bathe K.J., 1987, A finite element formulation for nonlinear incompressible hyperelastic and inelastic analysis, *Computers and Structures*, **26**, 357-409

18. Young J.M., Yao J., Ramasubramanian A., Taber L.A., Perucchio R., 2010, Automatic generation of user material subroutines for biomechanical growth analysis, *Journal of Biomechanical Engineering*, **132**, 10, doi: 10.1115/1.4002375

Manuscript received April 13, 2016; accepted for print January 16, 2017